DAA/Ames

JOINT INSTITUTE FOR AERONAUTICS AND ACOUSTICS
STANFORD UNIVERSITY

40 P.

IN-16838

JIAA TR-73

CALIBRATION AND DATA REDUCTION FOR A FIVE-HOLE PROBE

By

C. S. Lee and N. J. Wood

(JIAA-TR-73) CALIBRATION AND DATA REDUCTION FOR A FIVE-HOLE PROBE (Stanford Univ.) 40 p CSCL 20D

N86-29157

Unclas
H2/34 43335

Stanford University
Department of Aeronautics and Astronautics
Stanford, CA 94305

June 1986

JIAA TR-73

CALIBRATION AND DATA REDUCTION FOR A FIVE-HOLE PROBE

By

C. S. Lee and N. J. Wood

Stanford University
Department of Aeronautics and Astronautics
Stanford, CA 94305

June 1986

ABSTRACT

An experimental investigation was conducted to study a five-hole probe in low speed flow. The probe was a 0.125-in diameter tube with a hemisphere tip, the central hole was located at the tip of the probe, the four side holes were symmetrically located at 30° with respect to the probe axes. Calibrations were conducted in the Stanford low speed wind tunnel at two typical speeds. The data reduction was performed by double interpolations using spline curve fitting. The calibration and data reduction schemes are discussed and analyzed.

Results of the five-hole probe measurements in typical three-dimensional vortical flows, including velocity vectors in cross plane, pressure and vorticity contours are presented. Good repeatability is observed. The spatial resolution of the probe is found to be better than 0.05 inches.

NOMENCLATURE

b	wing span.
C_P	pressure coefficient.
C_μ	blowing ratio.
P	pressure.
$\overline{P}$	averaged pressure of the side holes.
q	dynamic pressure.
R	radius of the probe.
R_e	Reynolds number based on the probe tip diameter.
T	coordinate transformation matrix.
U	streamwise velocity.
V	spanwise velocity.
W	transverse velocity.
$\overline{U}$	magnitude of the velocity vector.
x	streamwise coordinate.
y	spanwise coordinate.
z	transverse coordinate.
α	pitch angle, angle of attack.
β	yaw angle.
ϕ	cone angle.
ψ	azimuth angle.
ξ	streamwise vorticity.

subscripts :

$_1$	central hole.
$_{2,3}$	side holes on the yaw plane.
$_{4,5}$	side holes on the pitch plane.
$_\infty$	free stream condition.

1. INTRODUCTION

The concept of vortical flow control to augment and stabilize the flow field around wings, such as tip blowing on a low aspect ratio wing, leading edge blowing or leading edge flaps on a delta wing, has been studied in several programs by the staff of the Joint Institute. The vortical flows in these studies can be characterized by strong vortices which induce favorable pressure distributions on the wing surfaces. Therefore, quantitative information on the strength and position of the vortices is essential to understand these flow fields.

For steady, three dimensional flows at low speeds, the independent variables necessary to describe the flow field are three components of velocity and static pressure. Five-hole probes provide a direct way to obtain all these quantities in a single measurement. However, this application requires (1) complete calibration which is unique to each probe, and (2) a complicated data reduction scheme to interpret the measured values. This report describes the calibration and data reduction procedures of a five-hole probe. Typical measured data, including velocity vectors, total and static pressures, and vorticity contours are shown. All the experiments were integrated with a minicomputer data acquisition system (PDP-11/23 with DECLAB-11/MNC system). The computer programs for calibration and data reduction are listed in the Appendices.

2. FIVE-HOLE PROBES

Five-hole probes, as the name implies, are probes with five pressure sensing holes. As shown in Fig. 1, a five-hole probe combines a total pressure tap (the central hole) with a yaw-head (two side holes on the horizontal plane) and a pitch-head (two side holes on the vertical plane). The idea is to obtain the magnitude of velocity from the dynamic pressure measured by the pressure difference between the central hole and the four side holes, and the direction of velocity from the pressure difference between the side holes in the pitch and yaw planes.

The probe employed in this study is a 0.125-in diameter tube with a hemisphere tip manufactured in the Department Machine Shop. The central hole was located at the center of the hemisphere, the four side holes were symmetrically located on the hemisphere at 30° with respect to the probe axes as shown in Fig.1. The probe tip was 2.0-in upstream of a 0.375-in diameter mounting tube to avoid the possible interference.

Huffman et al.[1] solved the small perturbation potential equation and obtained an analytical expression for surface pressure distribution on a slender body of revolution at angle of attack α, and side slip angle β. The sensitivity of surface pressure coefficient to the change in flow angles can then be derived

$$\frac{\partial(\Delta C_P)}{\partial\alpha} = 8R' \cos 2\alpha \cos^2 \beta, \tag{1}$$

where $\Delta C_P = C_{P90^\circ} - C_{P270^\circ}$, is the pressure difference of the pitch-head. R' is the rate change of radius in the probe axial direction.

This analytical equation is valuable as a guide to define functional coefficients for data reduction, but it can not replace the experimental calibration. The simplified assumptions (small perturbation, potential flow) and irregularity of the probes in manufacturing make it necessary to calibrate each probe individually. This equation shows that for a given probe geometry, the pressure sensitivity of pitch angle depends on both α and β. Therefore, a complete experimental calibration matrix at different combinations of α and β is needed due to the coupling between the pressure sensitivities and flow angles.

3. CALIBRATION

The objective of calibration is to experimentally determine a set of pressure data that can describe the probe's response to a known flow field. In application, total and static pressures, pitch, yaw angles, and the magnitude of velocity of the unknown flow field should be obtained from the measured pressures and these calibration data.

3.1. Transformation of Coordinates

Due to the constraint of the existing probe support mechanism, the angular position of the probe can only be changed through rotation of the probe about its own axis and rotation of the mounting tube. A coordinate transformation is therefore needed to relate these angles to the pitch and yaw angles of the flow.

As shown in Fig. 2, let the rotation of the probe be "azimuth angle", ψ, and the rotation of the mounting tube be "cone angle", ϕ. In the "azimuth-cone" calibration (the probe is rotated with an angle ψ about the x-axis, followed by a rotation of the mounting tube with an angle ϕ about the z-axis), the transformation matrix can be written as

$$T_1 = \begin{pmatrix} 1 & 0 & 0 \\ 0 & \cos\psi & -\sin\psi \\ 0 & \sin\psi & \cos\psi \end{pmatrix} \cdot \begin{pmatrix} \cos\phi & -\sin\phi & 0 \\ \sin\phi & \cos\phi & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

$$= \begin{pmatrix} \cos\phi & -\sin\phi & 0 \\ \cos\psi\sin\phi & \cos\psi\cos\phi & -\sin\psi \\ \sin\psi\sin\phi & \sin\psi\cos\phi & \cos\psi \end{pmatrix}.$$

Similarly, the transformation matrix for the "pitch-yaw" calibration (the probe is tilted forward or backward to provide a change in pitch angle, then rotated left or right by a yaw angle) is

$$T_2 = \begin{pmatrix} \cos\alpha & 0 & -\sin\alpha \\ 0 & 1 & 0 \\ \sin\alpha & 0 & \cos\alpha \end{pmatrix} \cdot \begin{pmatrix} \cos\beta & -\sin\beta & 0 \\ \sin\beta & \cos\beta & 0 \\ 0 & 0 & 1 \end{pmatrix}$$

$$= \begin{pmatrix} \cos\alpha\cos\beta & -\cos\alpha\sin\beta & -\sin\alpha \\ \sin\beta & \cos\beta & 0 \\ \sin\alpha\cos\beta & -\sin\alpha\sin\beta & \cos\alpha \end{pmatrix}.$$

During the probe calibration, the flow direction is aligned with x-axis, i.e. $(U, V, W) = (U_\infty, 0, 0)$ in the x-y-z coordinate. The velocity vector in the "azimuth-cone" coordinate is then

$$\begin{pmatrix} U_1 \\ V_1 \\ W_1 \end{pmatrix} = T_1 \cdot \begin{pmatrix} U_\infty \\ 0 \\ 0 \end{pmatrix} = U_\infty \begin{pmatrix} \cos\phi \\ \cos\psi\sin\phi \\ \sin\psi\sin\phi \end{pmatrix}. \tag{2}$$

Similarly, the velocity vector in the "pitch-yaw" coordinate is

$$\begin{pmatrix} U_2 \\ V_2 \\ W_2 \end{pmatrix} = T_2 \cdot \begin{pmatrix} U_\infty \\ 0 \\ 0 \end{pmatrix} = U_\infty \begin{pmatrix} \cos\alpha\cos\beta \\ \sin\beta \\ \sin\alpha\cos\beta \end{pmatrix}. \tag{3}$$

The velocity vector should be independent of the coordinates, so we can equate Eqs. 2 and 3

$$\cos\phi = \cos\alpha\cos\beta,$$

$$\cos\psi\sin\phi = \sin\beta,$$

$$\sin\psi\sin\phi = \sin\alpha\cos\beta.$$

Since we want to express the measured velocity vectors in terms of the pitch and yaw angles, the calibration matrix should consist of a set of predertermined α and β. In the actual experimental set-up, the angles are set by the azimuth and cone angles. The corresponding ϕ and ψ for given α and β can be obtained by rearranging the above equation, the corresponding ϕ and ψ for given α and β are

$$\psi = \tan^{-1}\left(\frac{\sin\alpha}{\tan\beta}\right),$$
$$\phi = \tan^{-1}\left(\frac{\tan\alpha}{\sin\psi}\right). \tag{4}$$

An easier equation for the cone angle is $\phi = \cos^{-1}(\cos\alpha\cos\beta)$, but arc-cosine cannot discriminate the positive and negative angles, therefore, ϕ is expressed as arc-tangent in Eq. 4.

3.2. Normalization of Pressures

A typical way to normalize pressures is to take the free stream dynamic pressure, q_∞, as the reference pressure. The dynamic head, which is independent of the probe condition, is a constant in the flow field, therefore the fluctuations in the measured pressure coefficients can not be smoothed by this constant. Using a local measured value as the normalization parameter is a better way to reduce the scatter in the experimental data[2]. A reasonable substitute would be the dynamic pressure at the measurement point, but this is not a quantity that can be measured directly from the pressure readings of the five-hole probe. Hence, the pressure difference of the central hole and the averaged value of the four side pressures is taken to normalize the pressure[3]. As shown in Fig. 1, the pressure coefficients are defined as

$$C_{Pyaw} = \frac{P_2 - P_3}{P_1 - \overline{P}},$$

$$C_{Ppitch} = \frac{P_4 - P_5}{P_1 - \overline{P}},$$

$$C_{Ptotal} = \frac{P_1 - P_{total}}{P_1 - \overline{P}}, \tag{5}$$

$$C_{Pstatic} = \frac{\overline{P} - P_{static}}{P_1 - \overline{P}},$$

$$\overline{P} = \frac{P_2 + P_3 + P_4 + P_5}{4}.$$

where P_{total} and P_{static} are the total and static pressures at the measurement point. Generally speaking, P_{total} and P_{static} are unknown quantities which depend on flow field. However, during the probe calibration in the wind tunnel, they are known and equal to the free stream values, so all the four pressure coefficients can be determined.

3.3. Calibration Results

The probe was calibrated in the Stanford low speed wind tunnel with nominal free stream velocities of 20 and 40 m/s corresponding to Reynolds numbers of 4.0×10^4 and 8.1×10^4 based on the probe tip diameter. The probe was set at one of the six predetermined yaw angles and swept in $10°$ increments through the pitch angle, the pressure coefficients were recorded at each calibration point. The calibration was repeated and it was found that the data was repeatable. The calibration program is listed in Appendix I.

The complete set of calibration data at $R_e = 8.1 \times 10^4$ is plotted in Figs. 3 to 6. Figs. 3 and 4 show the values of C_{Ppitch} and C_{Pyaw} at different α and β. At moderate yaw angles, $C_{P_{pitch}}$ is a linear function of pitch angle; likewise, at moderate pitch angles, $C_{P_{yaw}}$ only varies linearly with yaw angle. However, for the pitch and yaw angles larger than 15°, the pressure response is non-linear, and stronger coupling between pitch and yaw can be observed.

Three dimensional surface plots of the total and static pressure coefficients at different α and β are shown in Figs. 5 and 6. When $\alpha, \beta \approx 0$, the central hole measures the total pressure of the flow, therefore, $C_{P_{total}}$ has a maximum value of 0 at small α and β as expected. The stagnation point departs from the central hole as α and β increase, thereby decreasing $C_{P_{total}}$. $C_{P_{static}}$ is a measurement of the averaged surface pressure of the four side holes, hence it has a maximum when the probe is aligned with the flow. This maximum value depends on the probe geometry, for the present probe, the maximum $C_{P_{static}}$ is around 1.0. When the pitch and yaw angles increase, the pressure on the windward hole increases whereas the pressure on the leeward hole decreases, thereby decreasing $C_{P_{static}}$ as shown in Fig.6.

From these calibration curves, it can be observed that both total and static pressure coefficients vary with flow angles. Therefore in the application, all the unknown quantities including pitch and yaw angles, total and static pressures should be interpolated from these calibration curves.

4. DATA REDUCTION

For a set of measured pressure coefficients, pitch and yaw angles are determined from $C_{P_{pitch}}$ and $C_{P_{yaw}}$. After α and β are obtained, $C_{P_{total}}$ and $C_{P_{static}}$ can be determined from Figs. 5 and 6, therefore, the magnitude of the velocity can be calculated by the interpolated total and static pressures. Since these coefficients are coupled, a double interpolation is required. A data reduction procedure by two dimensional spline curve fitting is demonstrated in the following section.

Figs. 3 and 4 can be combined into one plot as $C_{P_{yaw}}$ versus $C_{P_{pitch}}$ for each calibration point of α and β as shown in Fig. 7-a. As an example, let the local measured pressure coefficients of an unknown velocity field be $C_{P_{pitch}} = 2.0$ and $C_{P_{yaw}} = 2.5$. A vertical line is drawn at $C_{P_{pitch}} = 2.0$, this line intercepts the constant yaw angle curves at six points as shown by the squares in Fig. 7-a. The corresponding $C_{P_{yaw}}$ values of these interceptions can be obtained from the ordinate. Consequently, the variation of $C_{P_{yaw}}$ versus the calibrated yaw angles for $C_{P_{pitch}} = 2.0$ can be plotted as shown in Fig. 7-b. This resulting monotonically increasing curve, when interpolated at $C_{P_{yaw}}$=2.5, yields the measured yaw angle ($\beta = 17.9°$ in this example). Similarly, the pitch angle can be obtained by drawing a horizontal line at $C_{P_{yaw}}$=2.5 in Fig. 7-a, and following the same procedure to construct the $C_{P_{pitch}} - \alpha$ curve, the resulting value is $\alpha = 18.5°$.

Knowing the local pitch and yaw angles, individual curves can be drawn to pass through the $C_{P_{static}} - \alpha$ data for different calibrated yaw angles as shown in Fig. 8-a. Then, the variation of $C_{P_{static}}$ versus β for $\alpha = 18.5°$ can be plotted and evaluated at $\beta = 17.9°$ to obtain the static pressure coefficient as shown in Fig. 8-b, which gives a resulting value of $C_{P_{static}} = 0.90$. $C_{P_{total}}$ can be evaluated in the same manner, and the resulting value is $C_{P_{total}} = -1.01$.

To check the dependency of the calibration on the Reynolds number, the above example is repeated by using the calibration data at $R_e = 4.0 \times 10^4$. The resultant values are $\alpha = 17.3°, \beta = 16.9°, C_{P_{total}} = -0.84$, and $C_{P_{static}} = 0.98$. The corresponding changes are $-6.5\%(\alpha), -5.6\%(\beta), -6.7\%(C_{P_{total}})$, and $-3.0\%(C_{P_{static}})$, these errors are within the acceptable measurement error in the present set-up, therefore, it is concluded that no significant changes in the pressure coefficients can be attributed to the Reynolds numbers.

These graphic procedures have been replaced by individual spline curve fitting through the calibrated data, and the adapted computer program is listed in Appendix II.

The magnitude of the velocity vector can be calculated by the Bernoulli's equation,

$$\overline{U} = \sqrt{\frac{2}{\rho}(P_{total} - P_{static})}, \tag{6}$$

where P_{total} and P_{static} can be expressed as functions of C_{Ptotal} and $C_{Pstatic}$ respectively. Rearranging Eq. 5

$$\begin{aligned} P_{total} &= P_1 - C_{Ptotal}(P_1 - \overline{P}), \\ P_{static} &= \overline{P} - C_{Pstatic}(P_1 - \overline{P}). \end{aligned} \tag{7}$$

Combining Eqs. 6 and 7, $\overline{U}$ can be calculated directly from the measured $P_1, \overline{P}$ and the interpolated values of C_{Ptotal} and $C_{Pstatic}$ as

$$\overline{U} = \sqrt{\frac{2}{\rho}(P_1 - \overline{P})(1 - C_{Ptotal} + C_{Pstatic})}. \tag{8}$$

After the magnitude and direction of the velocity vector are obtained, the three velocity components can be easily resolved by Eq. 3, i.e.

$$\begin{aligned} U &= \overline{U} \cos\alpha \cos\beta, \\ V &= \overline{U} \sin\beta, \\ W &= \overline{U} \sin\alpha \cos\beta. \end{aligned} \tag{9}$$

5. APPLICATIONS

As stated in the introduction, the objective of this study is application oriented, therefore, some results of flow field measurements in steady vortical flows are presented here to demonstrate its applicability.

A typical velocity vector plot in the cross plane is shown in Fig. 9 for a low aspect ratio wing with weak tip blowing[4]. The velocity vectors are plotted on the measuring station looking toward upstream. A well-defined tip vortex can be observed in this figure, and the shear layer resulting from the spanwise loading distribution can be noticed by the rapid change in velocity direction near the lower surface. The dash line represents the contour of the model projected on the measuring plane. The coordinates are normalized by the half span $b/2$.

Fig. 10 shows the total pressure contours for the same experimental condition, the pressure loss coefficient is defined as

$$\Delta C_{P_{total}} = \frac{P_{total} - P_{total\infty}}{q_\infty}. \tag{10}$$

The streamwise vorticity ξ can be obtained from the measured velocity components by

$$\xi = \frac{\partial W}{\partial y} - \frac{\partial V}{\partial z}. \tag{11}$$

Due to the relatively large spacing between the measurement points (0.25-in in this example), additional errors can be introduced if the data points are used directly to evaluate ξ. Therefore, spline curves were fitted to the measured cross flow velocities to generate velocity vectors in smaller grids, and the numerical differentiation were preformed by the central differencing scheme. The contour program is listed in Appendix III.

The resultant contour plot of vorticity is shown in Fig. 11 for the same case. By comparing Figs. 10 and 11, it can be observed that the total pressure contours are very similar to the vorticity contours. The regions of high total pressure loss almost coincide with the regions of high streamwise vorticity, but the vorticity contours are better defined and more symmetrical in the core region.

Fig. 12 shows the circumferential velocity distribution in the cross plane of a tip vortex for the same wing at $\alpha = 7^\circ$ without blowing. The different symbols represent different spacing between the measurement points. The data essentially follow a single curve, indicating the repeatability is very good. The velocity distribution near the core can be resolved by the smallest measurement spacing of 0.05 in, which is smaller than the probe diameter. The solid line represents the circumferential velocity distribution of a potential vortex of the corresponding strength. Except for the core region, the agreement between the measured data and the potential vortex is satisfactory on the right half of this figure. The deviation on the left half is due to the shear layer resulting from the wing.

6. CONCLUDING REMARKS

Procedures for the calibration of a five-hole probe have been developed and documented, and an efficient data reduction scheme has been demonstrated. No significant changes in the calibration data were found within the flow speeds of interested (20-40 m/s).

Automatic flow mapping programs have been interfaced with the existing mini-computer system. The probe was employed to survey the three dimensional flow fileds of wing tip vortices. Three components of velocity vector, total and static pressures were readily obtained from the measurements. Quantitative values of the vorticity were derived from the measured data. The present technique was shown to be able to resolve detailed quantities in vortical flows.

REFERENCES

1. Huffman, G. D., Rabe, D. C., and Poti, N. D., "Flow Direction Probes from a Theoretical Point of View". J. Phys. E: Sci. Instrum., pp. 751-760, Vol. 13, 1980.

2. Krause, L. N., and Dudzinski, T. J., "Flow-Direction Measurement with Fixed Position Probes in Subsonic Flows over a Range of Reynolds Number", NASA TMX-52576, 1969.

3. Treaster, A. L., and Yocum, A. M., "The Calibration and Application of Five-hole Probes". Instrum. Society of America Transactions, pp. 23-34, Vol. 18, No. 3, 1979.

4. Lee, C. S., Tavella, D., Wood, N. J., and Roberts, L., "Flow Structure of Lateral Wing-Tip Blowing". AIAA paper 86-1810, 1986.



Figure 1. Schematic of the five-hole probe and velocity resolution.



Figure 2. The "azimuth-cone" transformation.



Figure 3. Typical calibration curves for $C_{P_{pitch}}$ at different α and β.



Figure 4. Typical calibration curves for $C_{P_{yaw}}$ at different α and β.



Figure 5. Typical calibration curves for $C_{P_{total}}$ at different α and β.



Figure 6. Typical calibration curves for $C_{P_{static}}$ at different α and β.



Figure 7. Graphical procedures to obtain β.



Figure 8. Graphical procedures to obtain $C_{P_{static}}$.



Figure 9. Typical velocity vector plot of a tip vortex in the cross plane.



Figure 10. Total pressure contours. (unit in % $\triangle C_{pT}$)



Figure 11. **Vorticity contours. (unit in $\xi b/2u_\infty$)**



Figure 12. Circumferential velocity distribution of a tip vortex.

APPENDIX I. The Probe Calibration Program

ORIGINAL PAGE IS
OF POOR QUALITY

```
C
C
C
      PROGRAM P5CALB
C   JOHNSON LEE, JUL-09-85
C   5-HOLE PROBE CALIBRATION PROGRAM
      DIMENSION V0(5),VC(5),VM(5),P(5),ALFA(6),BETA(6)
      DIMENSION CPYAW(6,6),CPPCH(6,6),CPTOT(6,6),CPSTA(6,6)
C   OPEN CALIBRATION DATA FILE, THE FILE NAME IS GIVEN
      CALL OPENFL(15,0)
C   SET TUNNEL CONDITIONS.
      TYPE *,' ENTER TOTAL TEMPERATURE (deg.C)?'
      ACCEPT *, TTOT
      TYPE *,' ENTER AMBIENT PRESSURE (in.HG)?'
      ACCEPT *, PBARO
C*** Patm in lbf/ft.sq, Rtot in deg.R
      PATM=847.*PBARO/12.
      RTOT=1.8*TTOT+492.
C   CACULATE DENSITY FROM THE IDEAL GAS LAW
      RHO=PATM/53.3/RTOT
C*** Rho IN Lbm/ft.cu; NOW, 1(Lbm/Ft.cu)=16.052(Kg/m.cu)
      RHO=RHO*16.052
C   ASK FOR DESIRED Uinf?
      TYPE *,' ENTER UINF (M/SEC)?'
      ACCEPT *, UINF
C*** From tunnel calibration: q = 1.63 + 1201.5*h
C***  h("H2O across taps), q(Nwtn/m.sq)
      DP=(.5*RHO*UINF**2-1.63)/1201.5
      TYPE *,' DP(DYNAMIC PRESSURE) TO BE SET AT ',DP,'  IN.H2O'
      TYPE *,' ENTER DP (IN.WATER)?'
      ACCEPT *,DP
      UINF=SQRT(2.*(1.63+1201.5*DP)/RHO)
C*** Probe diameter : 0.125in; 1(in)=0.0254(m)
      DIA=0.125*0.0254
C*** Air viscosity mu=1.83e-6 (kg.sec/m.sq)
      GMU=1.83E-6
      RE=DIA*UINF*RHO/GMU
      TYPE 910, TTOT,PBARO,DP,UINF,RE
910   FORMAT(///5X,'AMBIENT TEMP=',F5.1,' DEG.C.'/
     *   5X,'AMBIENT PRESSURE =',F7.3,'  IN.HG.'/
     *   5X,'DP(DYNAMIC PRESSURE) =',F7.3,'  IN.WATER'/
     *   10X,'FREE STREAM VEL =',F7.3,'  M/S'/
     *   10X,'REYNOLDS NO. =',F9.0///)
      PAUSE ' OPEN THE 5-TH HOLE'
C calibrate the transducer sensitivity
C scanivalve 'home' opens to atmosphere
      CALL HOME
C  Five transducers: from A/D ch.15 to ch.19, Take zero pressure reading
      CALL ADMCH(15,5,V0)
C  Set calibr. pressure at 0.5 psi on scannivalve ch.5-8
      CALL STEP
      PAUSE 'SET CALIBR. PRESSURE AT 0.5 psi(CLOCKWISE), <CR> TO START.'
      CALL ADMCH(15,4,VC)
      PAUSE 'SET CALIBR. PRESS. FOR THE 5-TH HOLE, <CR> TO START'
      CALL ADMCH(19,1,VC(5))
      CALL STEP
      PAUSE 'CONNET 5-TH HOLE.'
C  Taketunnel total & static pressures, Ch.9 & 10.
      CALL ADMCH(15,2,VM)
C  Convert voltage into psi
```

```
      PTOT=0.5*(VM(1)-V0(1))/(VC(1)-V0(1))
      PSTA=0.5*(VM(2)-V0(2))/(VC(2)-V0(2))
      CALL STEP
C  Now, ready to take calibr. data, calibration in the pitch-yaw coordinate,
C        #1: Center, #2: Right, #3: Left, #4: Lower, #5: Upper
      DELTA=10.
      DO 100 I=1,6
      DO 100 J=1,6
      BETA(I)=-25.+(I-1)*DELTA
      ALFA(J)=-25.+(J-1)*DELTA
      TYPE 900,ALFA(J),BETA(I)
900   FORMAT(//2X,' Set Alfa =',F5.1,' Beta =',F5.1,5X,'<CR> to start')
      PAUSE

C  subroutine ADMCH(starting channel no., no. of channels, output voltage)
      CALL ADMCH(15,5,VM)
      DO 110 K=1,5
110   P(K)=0.5*(VM(K)-V0(K))/(VC(K)-V0(K))
      PBAR=0.25*(P(2)+P(3)+P(4)+P(5))
      DENOM=1./(P(1)-PBAR)
      CPYAW(I,J)=(P(2)-P(3))*DENOM
      CPPCH(I,J)=(P(4)-P(5))*DENOM
      CPTOT(I,J)=(P(1)-PTOT)*DENOM
      CPSTA(I,J)=(PBAR-PSTA)*DENOM
      TYPE *,'Cyaw,Cpit,Ctot,Csta=',CPYAW(I,J),CPPCH(I,J),CPTOT(I,J),
     *          CPSTA(I,J)
100   CONTINUE
C   STORE DATA IN THE DATA FILE (P5CALB.DAT)
      WRITE(15,901) RE,UINF,TTOT,PBARO,DP,PSTA,PTOT
      DO 200 I=1,6
      DO 200 J=1,6
200   WRITE(15,902) BETA(I),ALFA(J),CPYAW(I,J),CPPCH(I,J),CPTOT(I,J)
     *  ,CPSTA(I,J)
      CLOSE(UNIT=15,DISPOSE='SAVE')
901   FORMAT(E10.2,6F7.2)
902   FORMAT(2F6.1,5F9.4)
      STOP
      END
```

APPENDIX II. The Flow Mapping Program

```
C
C
C
        PROGRAM P5MAPP
C   JOHNSON LEE, JUL-09-85
C   5-HOLE PROBE FLOW MAPPING PROGRAM
C
C     THIS PROGRAM DOES THE AUTOMATIC PROBE TRAVERING IN THE CORSS
C      PLANE, AND TAKE DATA AT EACH GRID POINT.
C
C    LINK P5MAPP,OPENFL,SPLINE,P5MNA,SY:MNCSNG,SY:TRALIB
C
        DIMENSION VM(5),P(5),ALFA(6),BETA(6)
        DIMENSION CPYAW(6,6),CPPCH(6,6),CPTOT(6,6),CPSTA(6,6)
        DIMENSION SCYP(6,6),ELYP(6,6),ELPY(6,6),SCPY(6,6)
        DIMENSION SCSP(6,6),ELSP(6,6),SCTP(6,6),ELTP(6,6)
        DIMENSION X(6),Y(6),YY(6),SC(6),EL(6),DA(6),DB(6),DC(6)
        NC=6
        RAD=3.14159/180.
C   SET TUNNEL CONDITIONS.
        TYPE *,' ENTER TOTAL TEMPERATURE (deg.C)?'
        ACCEPT *, TTOT
        TYPE *,' ENTER AMBIENT PRESSURE (in.HG)?'
        ACCEPT *, PBARO
C*** Patm in lbf/ft.sq, Rtot in deg.R
        PATM=847.*PBARO/12.
        RTOT=1.8*TTOT+492.
C   CACULATE DENSITY FROM THE IDEAL GAS LAW
        RHO=PATM/53.3/RTOT
C*** Rho IN Lbm/ft.cu; NOW, 1(Lbm/Ft.cu)=16.052(Kg/m.cu)
        RHO=RHO*16.052
C   ASK FOR DESIRED Uinf?
        TYPE *,' ENTER UINF (M/SEC)?'
        ACCEPT *, UINF
C*** From tunnel calibration: q = 1.63 + 1201.5*h
C***  h:("H2O across taps), q:(Nwtn/m.sq)
        DP=(.5*RHO*UINF**2-1.63)/1201.5
        TYPE *,' DP(DYNAMIC PRESSURE) TO BE SET AT ',DP,'  IN.H2O'
        TYPE *,' ENTER DP (IN.WATER)?'
        ACCEPT *,DP
        UINF=SQRT(2.*(1201.5*DP+1.63)/RHO)
        TYPE 910, TTOT,PBARO,DP,UINF
910     FORMAT(///14X,'AMBIENT TEMP=',F6.2,' DEG.C.'/
     *    9X,'AMBIENT PRESSURE =',F6.2,' IN.HG.'/
     *    5X,'DP(DYNAMIC PRESSURE) =',F6.2,' IN.WATER'/
     *    10X,'FREE STREAM VEL =',F6.2,' M/S'//)
C
        TYPE *,' ENTER ANGLE OF ATTACK (DEG.)?'
        ACCEPT *,ATTK
C   THE BLOWING RATIO,
        TYPE *,' ENTER CMU?'
        ACCEPT *,CMU
        TYPE *,' ENTER MEASURED PLANE, Xt/C?'
        ACCEPT *,XOC
C   OPEN DATA FILE TO BE STORED,
        TYPE *,' OPEN THE DATA FILE TO STORED THE VEL. FIELD,'
        CALL OPENFL(25,0)
C   READ THE OLD CALIBRATION DATA FILE(P5CLB1.DAT),
        OPEN(UNIT=15,NAME='P5CLB1.DAT',TYPE='OLD')
        READ(15,*)
```

```
        DO 100 I=1,NC
        DO 100 J=1,NC
100     READ(15,*) BETA(I),ALFA(J),CPYAW(I,J),CPPCH(I,J),CPTOT(I,J)
     *   ,CPSTA(I,J)
        CLOSE(UNIT=15,DISPOSE='SAVE')
C   CROSS SPLINE CURVE FIT FOR THE CALIBRATION DATA
        DO 110 I=1,NC
        CALL EX12D(I,NC,X,CPPCH,1,1)
        CALL EX12D(I,NC,Y,CPYAW,1,1)
        CALL SPFIT(NC,X,Y,SC,EL,DA,DB,DC)
        CALL EX12D(I,NC,SC,SCYP,0,1)
        CALL EX12D(I,NC,EL,ELYP,0,1)
110     CONTINUE
        DO 120 J=1,NC
        CALL EX12D(J,NC,X,CPYAW,1,0)
        CALL EX12D(J,NC,Y,CPPCH,1,0)
        CALL SPFIT(NC,X,Y,SC,EL,DA,DB,DC)
        CALL EX12D(J,NC,EL,ELPY,0,0)
        CALL EX12D(J,NC,SC,SCPY,0,0)
120     CONTINUE
        DO 130 I=1,NC
        CALL EX12D(I,NC,Y,CPSTA,1,1)
        CALL SPFIT(NC,ALFA,Y,SC,EL,DA,DB,DC)
        CALL EX12D(I,NC,SC,SCSP,0,1)
        CALL EX12D(I,NC,EL,ELSP,0,1)
130     CONTINUE
        DO 140 I=1,NC
        CALL EX12D(I,NC,Y,CPTOT,1,1)
        CALL SPFIT(NC,ALFA,Y,SC,EL,DA,DB,DC)
        CALL EX12D(I,NC,SC,SCTP,0,1)
        CALL EX12D(I,NC,EL,ELTP,0,1)
140     CONTINUE
C  Set the cannivalve to home,
        CALL HOME
C  Take tunnel total & static pressures, Ch.1 & 2.
        CALL ADMCH(8,2,VM)
C  Convert voltage into psi, sensitivity: 0.1 psi/volt
        PTOT=0.1*VM(1)+PATM/144.
        PSTA=0.1*VM(2)+PATM/144.
        WRITE(25,901) ATTK,CMU,XOC,UINF,DP,PSTA,PTOT
901     FORMAT(X,9F8.2)
C
C   Now, Ready to take the measurement
C
        TYPE *,' ENTER THE INITIAL POSITION, (Y0,Z0)?'
        ACCEPT *,YM,ZM
        CALL TRINIT
C   Origin is located at the lower-right corner when facing the flow,
500     TYPE *,' ENTER THE TOTAL VERTICAL DISTANCE (INCH)?'
        ACCEPT *,ZTOT
        TYPE *,' ENTER INCREMENT IN THE VERTICAL DIRECTION?'
        ACCEPT *,DZ
        TYPE *,' ENTER THE TOTAL HORIZONTAL DISTANCE (INCH)?'
        ACCEPT *,YTOT
        TYPE *,' ENTER INCREMENT IN THE HORIZONTAL DIR.?'
        ACCEPT *,DY
        NZ=ZTOT/DZ+1
        NY=YTOT/DY+1
C   Get rid of the round-off error,
        DY=DY+0.0000001
```

```
      DZ=DZ+0.0000001
      TYPE *,' '
      TYPE *,' Z AXIS: Ztot, Dz=',ZTOT,DZ
      TYPE *,' Y AXIS: Ytot, Dy=',YTOT,DY
      TYPE *,' '
      TYPE *,' ENTER 0, IF THIS IS O.K.!'
      ACCEPT *,IFF
      IF(IFF.NE.0) GO TO 500
      WRITE(25,902) NZ,NY
902   FORMAT(2I4)
C  5-Hole Probe: From scannivalve ch.5-8,; Center hole: the 5-th transducer
C        #1: Center, #2: Right, #3: Left, #4: Lower, #5: Upper
C
C  the 5-hole probe is connected at the 2nd scannivalve channel,
      CALL STEP
      DO 510 IH=1,NY
      DO 530 IV=1,NZ
      CALL ADMCH(8,5,VM)
      DO 200 K=1,5
200   P(K)=0.1*VM(K)
      PBAR=0.25*(P(2)+P(3)+P(4)+P(5))
      DENOM=1./(P(1)-PBAR)
      CPY=(P(2)-P(3))*DENOM
      CPP=(P(4)-P(5))*DENOM
C  Interpolate for Alfa, Beta, CPsta & CPtot
      DO 210 I=1,NC
      CALL EX12D(I,NC,X,CPPCH,1,1)
      CALL EX12D(I,NC,Y,CPYAW,1,1)
      CALL EX12D(I,NC,EL,ELYP,1,1)
      CALL EX12D(I,NC,SC,SCYP,1,1)
210   CALL SPGET(NC,X,Y,SC,EL,CPP,YY(I),YP,YDP)
      CALL SPFIT(NC,YY,BETA,SC,EL,DA,DB,DC)
      CALL SPGET(NC,YY,BETA,SC,EL,CPY,YAW,YP,YDP)
C
      DO 220 J=1,NC
      CALL EX12D(J,NC,X,CPYAW,1,0)
      CALL EX12D(J,NC,Y,CPPCH,1,0)
      CALL EX12D(J,NC,SC,SCPY,1,0)
      CALL EX12D(J,NC,EL,ELPY,1,0)
220   CALL SPGET(NC,X,Y,SC,EL,CPY,YY(J),YP,YDP)
      CALL SPFIT(NC,YY,ALFA,SC,EL,DA,DB,DC)
      CALL SPGET(NC,YY,ALFA,SC,EL,CPP,PITCH,YP,YDP)
C
      DO 230 I=1,NC
      CALL EX12D(I,NC,Y,CPSTA,1,1)
      CALL EX12D(I,NC,SC,SCSP,1,1)
      CALL EX12D(I,NC,EL,ELSP,1,1)
230   CALL SPGET(NC,ALFA,Y,SC,EL,PITCH,YY(I),YP,YDP)
      CALL SPFIT(NC,BETA,YY,SC,EL,DA,DB,DC)
      CALL SPGET(NC,BETA,YY,SC,EL,YAW,CPS,YP,YDP)
C
      DO 240 I=1,NC
      CALL EX12D(I,NC,Y,CPTOT,1,1)
      CALL EX12D(I,NC,SC,SCTP,1,1)
      CALL EX12D(I,NC,EL,ELTP,1,1)
240   CALL SPGET(NC,ALFA,Y,SC,EL,PITCH,YY(I),YP,YDP)
      CALL SPFIT(NC,BETA,YY,SC,EL,DA,DB,DC)
      CALL SPGET(NC,BETA,YY,SC,EL,YAW,CPT,YP,YDP)
C   Finally, get the pitch, yaw, Cps, and Cpt!
C
```

```
C***  1.(Nt/m.sq)=0.0001451 (psi)
        DELTP=(P(1)-PBAR)*(1.+CPS-CPT)/0.0001451
        IF(DELTP.LE.0.) DELTP=0.
        VEL=SQRT(2.*DELTP/RHO)
        VX=VEL*COS(PITCH*RAD)*COS(YAW*RAD)
        VZ=VEL*SIN(PITCH*RAD)*COS(YAW*RAD)
        VY=VEL*SIN(YAW*RAD)
        TYPE 800,YM,ZM,VEL,PITCH,YAW,VX,VY,VZ
800     FORMAT(//25X,' (y,z) = (',2F8.2,')'/
     *          4X,'Vel(m/s), Pitch & Yaw (deg) =',3F8.3/
     *          12X,' Vx, Vy, Vz (m/sec) =',3F8.3)
        WRITE(25,903) VEL,PITCH,YAW,CPS,CPT
903     FORMAT(2(F5.2,X),2(F7.3,X),F7.3)
                IF(IV.EQ.NZ) GO TO 530
C    the probe is moving up and down in alternate sweeps to save time,
                DDZ=(-1)**(IH+1)*DZ
C    first, traverse the probe in Z-axis at fixed Y,
                CALL RMOVEZ(DDZ)
                ZZ=ZZ+DDZ
                ITIC=24*DZ
C    wait a little while for the probe to get the position
                CALL ISLEEP(0,0,0,ITIC)
                ZM=ZM+DDZ
530        CONTINUE
        IF(IH.EQ.NY) GO TO 510
C   now, the probe is moving in the Y-axis,
        CALL RMOVEY(DY)
        YM=YM+DY
        ITIC=24*DY
510     CALL ISLEEP(0,0,0,ITIC)
C   get the probe back to the original position,
        CALL RMOVEY(-YTOT)
        ITIC=24*YTOT
        CALL ISLEEP(0,0,0,ITIC)
        YM=YM-YTOT
        DDZ=-(-1)**IH*ZTOT
        CALL RMOVEZ(DDZ)
        ITIC=24*ZTOT
        CALL ISLEEP(0,0,0,ITIC)
        ZM=ZM+DDZ
        CLOSE(UNIT=25)
        CALL TRESCP
        STOP
        END
C
        SUBROUTINE EX12D(JK,N,AR1,AR2,IO,KORJ)
C
C   THIS SUBROUTINE INTERCHANGES AN ARRAY TO AND FROM 1-D & 2-D
C    JK: the fixed index in the 2-D array,
C     N: the no. of data in the 1-D array,
C   AR1: the 1-D array of N,
C   AR2: the 2-D array of NxN,
C    IO: 0: put 1-D data into 2-D array, 1: put 1 row of 2-D data into 1-d array
C  KORJ: 0: the fixed index is the 2nd one, i.e. change 1 row in the 2-D data;
C        1: the fixed index is the 1st one, i.e. change 1 column in the 2-D data.
C
        DIMENSION AR1(6),AR2(6,6)
        IF(KORJ) 40,40,1
1       IF(IO)   20,20,2
2       DO 10 K=1,N
```

```
10      AR1(K)=AR2(JK,K)
        RETURN
20      DO 30 K=1,N
30      AR2(JK,K)=AR1(K)
        RETURN
40      IF(IO)   70,70,50
50      DO 60 J=1,N
60      AR1(J)=AR2(J,JK)
        RETURN
70      DO 80 J=1,N
80      AR2(J,JK)=AR1(J)
        RETURN
        END
C
        SUBROUTINE SPFIT(N,X,Y,G,DX,A,B,C)
C
C   This subroutine fits a spline curve through a set of given data points,
C
C    input: N: no. of data points,
C           X: the independent data array,
C           Y: the dependent data array,
C    output: G: the array which holds the spline curve coefficients,
C           DX: the segment length array for X,
C    dummy working space: A,B,C, all have the dimension of N.
C
C     Reference: "Best Approximation Properties of the Spline Fit".
C                Walsh, J.L., etc, Journal of Mathematics and Mechanics,
C                        Vol. II, No. 2, pp. 225-234, 1962.
C
        DIMENSION X(1),Y(1),DX(1),A(1),B(1),C(1),G(1)
        DO 1 I=1,N-1
1       DX(I+1)=X(I+1)-X(I)
        DO 2 I=1,N-2
        G(I)=(Y(I+2)-Y(I+1))/DX(I+2)-(Y(I+1)-Y(I))/DX(I+1)
        B(I)=(DX(I+1)+DX(I+2))/3.
        A(I)=DX(I+1)/6.
2       C(I)=DX(I+2)/6.
        C(1)=C(1)/B(1)
        DO 3 I=2,N-2
        B(I)=B(I)-A(I)*C(I-1)
3       C(I)=C(I)/B(I)
        B(1)=G(1)/B(1)
        DO 4 I=2,N-2
4       B(I)=(G(I)-A(I)*B(I-1))/B(I)
        G(N-2)=B(N-2)
        DO 5 I=1,N-3
        K=N-2-I
5       G(K)=B(K)-C(K)*G(K+1)
        DO 6 I=1,N-2
        K=N-I
6       G(K)=G(K-1)
        G(1)=0.
        G(N)=0.
        RETURN
        END
C
        SUBROUTINE SPGET(N,X,Y,G,DX,AX,AY,AYP,AYDP)
C
C    This subroutine generates the dependent value AY for a given AX,
C
```

```
C      input: N: no. of data points ,
C              X: the independent variable array,
C              Y: the dependent varaible array,
C              G: the spline curve coefficients from SPFIT,
C             DX: the segment length array from subroutine SPFIT,
C             AY: the independent variable to be interpolated,
C
C      output: AY: the corresponding dependent variable for AX,
C              AYP: the first derivative at AY, i.e. dy/dx,
C             AYDP: the second derivative at AY, i.e. y".
C
       DIMENSION X(1),Y(1),G(1),DX(1)
       K=2
84     IF(AX-X(K)) 83,83,85
85     IF(AX-X(N)) 87,87,88
87     K=K+1
       GO TO 84
88     K=N
83     KK=K-1
       FF=G(KK)*(X(K)-AX)**3/(6.*DX(K))+G(K)*(AX-X(KK))**3/(6.*DX(K))
       AY=FF+(Y(K)/DX(K)-G(K)*DX(K)/6.)*(AX-X(KK))
       AY=AY+(Y(KK)/DX(K)-G(KK)*DX(K)/6.)*(X(K)-AX)
       BF=G(K)*(AX-X(KK))**2/(2.*DX(K))-G(KK)*(X(K)-AX)**2/(2.*DX(K))
       AYP=BF+(Y(K)-Y(KK))/DX(K)-DX(K)*(G(K)-G(KK))/6.
       AYDP=G(KK)*(X(K)-AX)/DX(K)+G(K)*(AX-X(KK))/DX(K)
       RETURN
       END
```

APPENDIX III. The Vorticity Contour Program

```
C
C
C
C   JOHNSON LEE, MAR-13-86
C
C     THIS PROGRAM PLOT THE VORTICITY CONTOURS ON THE CROSS FLOW, AND
C      INTEGRATE THE VORTICITY INSIDE A GIVEN CONTOUR TO OBTAIN THE
C      TOTAL CIRCULATION.  ALL THE SUBROUTINES ARE IN THE NASA AMES
C      VAX-11 LIBRARY (DISSPLA, MAP, ETC...)
C
        LOGICAL*1 NFILE(10)
        COMPLEX ZBL(8),ZZ
        DIMENSION PX(21),PY(21),SCPT(21,21),ELPT(21,21),VORT(31,31)
        DIMENSION XY(31,31,2),Q(31,31,2),VY(21,21),VZ(21,21)
        DIMENSION YY(21),YYY(21),SC(21),EL(21),D1(21),D2(21),D3(21)
        DIMENSION RWK(4000),IWK(60),XGRID(31),YGRID(31),PLEV(40)
        DATA NFILE/6*'0','.','D','A','T'/
C
        RAD=3.1416/180.
C   OPEN DATA FILE TO BE READ.
        TYPE *,' ENTER THE DATA FILE NAME?'
        ACCEPT 80,(NFILE(J),J=1,6)
        OPEN(UNIT=25,FILE=NFILE,STATUS='OLD')
        READ(25,*) ATTK,A,A,UINF,PATM,PSINF,PTINF
        READ(25,*) Z0,Y0,ZTOT,YTOT,DZ,DY
        ZM=Z0
        YM=Y0
        NZ=ZTOT/DZ+1
        NY=YTOT/DY+1
        DO 100 IY=1,NY
        DO 110 IZ=1,NZ
        READ(25,*) VEL,PITCH,YAW,DPT,DPS
C
        IZZ=IZ
        IF((IY/2*2-IY).EQ.0) IZZ=NZ-IZ+1
        IYY=IY
C   REVERSE LEFT AND RIGHT IF ATTK>0.
        IF(ATTK.LT.0.) GO TO 111
        IYY=NY-IY+1
        YAW=-YAW
111     VY(IZZ,IYY)=VEL*SIN(PITCH*RAD)*COS(YAW*RAD)/UINF
        VZ(IZZ,IYY)=VEL*SIN(YAW*RAD)/UINF
C
        IF(IY.EQ.1) PX(IZ)=ZM/9.
        ZM=ZM+DZ
110     CONTINUE
        IF(ATTK.GT.0.) THEN
        PY(IYY)=1.-YM/9.
        ELSE
        PY(IYY)=YM/9.-1.
        ENDIF
        YM=YM+DY
100     CONTINUE
        CLOSE(UNIT=25)
C
        TYPE *,' ENTER THE GRID SIZE? (NX and NY)?'
        ACCEPT *,NGX,NGY
        DGX=(PX(NZ)-PX(1))/(NGX-1)
        DGY=(PY(NY)-PY(1))/(NGY-1)
C
```

```
C   SPLINE CURVE INTERPOLATION TO REFINE THE DATA SET
        DO 133 IY=1,NY
        CALL EX12D(NZ,YY,IY,VY,1,0)
        CALL SPFIT(NZ,PX,YY,SC,EL,D1,D2,D3)
        CALL EX12D(NZ,SC,IY,SCPT,0,0)
133     CALL EX12D(NZ,EL,IY,ELPT,0,0)
C
        DO 130 I=1,NGX
        DO 130 J=1,NGY
        CYY=PY(1)+DGY*(J-1)
        CXX=PX(1)+DGX*(I-1)
C  INTERPOLATE FOR VY,
        DO 135 IY=1,NY
        CALL EX12D(NZ,YY,IY,VY,1,0)
        CALL EX12D(NZ,SC,IY,SCPT,1,0)
        CALL EX12D(NZ,EL,IY,ELPT,1,0)
135     CALL SPGET(NZ,PX,YY,SC,EL,CXX,YYY(IY),YP,YDP)
        CALL SPFIT(NY,PY,YYY,SC,EL,D1,D2,D3)
        CALL SPGET(NY,PY,YYY,SC,EL,CYY,CLV,YP,YDP)
C
        XY(I,J,1)=CXX
        XY(I,J,2)=CYY
        Q(I,J,1)=CLV
130     CONTINUE
C
        DO 143 IY=1,NY
        CALL EX12D(NZ,YY,IY,VZ,1,0)
        CALL SPFIT(NZ,PX,YY,SC,EL,D1,D2,D3)
        CALL EX12D(NZ,SC,IY,SCPT,0,0)
143     CALL EX12D(NZ,EL,IY,ELPT,0,0)
C
        DO 140 I=1,NGX
        DO 140 J=1,NGY
        CYY=PY(1)+DGY*(J-1)
        CXX=PX(1)+DGX*(I-1)
C   INTERPOLATE FOR VZ,
        DO 145 IY=1,NY
        CALL EX12D(NZ,YY,IY,VZ,1,0)
        CALL EX12D(NZ,SC,IY,SCPT,1,0)
        CALL EX12D(NZ,EL,IY,ELPT,1,0)
145     CALL SPGET(NZ,PX,YY,SC,EL,CXX,YYY(IY),YP,YDP)
        CALL SPFIT(NY,PY,YYY,SC,EL,D1,D2,D3)
        CALL SPGET(NY,PY,YYY,SC,EL,CYY,CLV,YP,YDP)
C
        Q(I,J,2)=CLV
140     CONTINUE
C
        CALL VORTIC(NGX,NGY,XY,Q,VORT)
C
C   NOW READY FOR THE CONTOUR PLOT,
C    ALL THE SUBROUTINES HERE ARE ADOPTED FROM THE "DISSPLA" PLOTTING
C     PACKAGE.
C
        CALL DIP(58,'PLOT.DIP',8)
        CALL PAGE(11.,8.5)
        CALL NOBRDR
        CALL HWSCAL('SCREEN')
        CALL PHYSOR(1.5,1.5)
        CALL NOCHEK
        CALL GRACE(0.0)
```

```
      CALL AREA2D(6.,6.)
      TYPE *,' ENTER THE PLOT AXIES: Xmin, Xmax, Ymin, Ymax?'
      ACCEPT *,XMIN,XMAX,YMIN,YMAX
      CALL GRAF(XMIN,0.25,XMAX,YMIN,0.25,YMAX)
C   SEARCH FOR THE MAX. AND MIN. VORT. VALUES,
      VMAX=-100000.
      VMIN= 100000.
      DO 230 J=1,NGX
      DO 230 I=1,NGY
      XGRID(I)=XY(I,J,1)
      YGRID(J)=XY(I,J,2)
      IF(VORT(I,J).GT.VMAX) VMAX=VORT(I,J)
      IF(VORT(I,J).LT.VMIN) VMIN=VORT(I,J)
230   CONTINUE
      TYPE *,' Vmax, Vmin =',VMAX,VMIN
C  ASK FOR CONTOUR LEVELS TO BE PLOTTED,
      TYPE *,' ENTER NO. OF CONTOUR LEVELS TO BE PLOTTED?'
      ACCEPT *,NLINE
      DVORT=(VMAX-VMIN)/(NLINE-1)
      DO 240 I=1,NLINE
      PLEV(I)=VMIN+(I-1)*DVORT
240   TYPE *,' I =',I,' VORT =',PLEV(I)
      CALL DASH
      ISIZ1=1000
      ISIZ2=20
      IXCON=1
      IYCON=1+ISIZ1
      ICK=ISIZ1+IYCON
      INAD=1
      ILEV=ISIZ2+1
      INPL=ISIZ2+ILEV
      DO 250 I=1,NLINE
      IWK(INAD)=0
      CALL MAP(PSI,NGX ,RWK(IXCON),RWK(IYCON),1,2,PLEV(I),
     +         IWK(INAD),ISIZ1,ISIZ2,RWK(ICK),1,1,NGX,NGY,
     +         IWK(ILEV),XGRID,YGRID,1)
      NLINS=IWK(INAD)
      IWK(NAD)=1
      DO 250 LINE=1,NLINS
      I1=IWK(INAD-1+LINE)
      NPTS=IWK(INAD+LINE)-I1
      IX=IXCON-1+I1
      IY=IYCON-1+I1
      CALL CURVE(RWK(IX),RWK(IY),NPTS,0)
250   CONTINUE
      CALL RESET('ALL')
      CALL ENDPL(0)
      CALL DONEPL
C
C  NOW, READY FOR THE CIRCULATION INTEGRATION,
C
301   TYPE *,' ENTER: 0 IF WANT TO INTEGRATE FOR CIRCULATION?'
      ACCEPT *,IFF
      IF(IFF.NE.0) GO TO 999
      TYPE *,' ENTER NO. OF CORNERS? (0 TO STOP)'
      ACCEPT *,NBL
      DO 310 K=1,NBL
      TYPE *,' ENTER COUNTER-CLOCKWISED PT (X,Y) FOR #',K
      ACCEPT *,ZX,ZY
310   ZBL(K)=CMPLX(ZX,ZY)
```

```
C   INTEGRATE VORTICITY INSIDE THE POLYGON,
        SUM=0.
        DO 350 I=1,NGX
        DO 350 J=1,NGY
        ZZ=CMPLX(XY(I,J,1),XY(I,J,2))
        DO 360 KK=1,NBL
        KKP=1+MOD(KK,NBL)
C   REJECT THE POINT WHICH IS OUTSIDE THE POLYGON,
        IF(AIMAG((ZZ-ZBL(KK))*CONJG(ZBL(KKP)-ZBL(KK))).LT.0.)GO TO 350
360     CONTINUE
        SUM=SUM+VORT(I,J)*DGX*DGY
350     CONTINUE
        TYPE *,' GAMMA=',SUM
        GO TO 301
80      FORMAT(6A1)
999     STOP
        END
C
      SUBROUTINE VORTIC(IDIM,JDIM,XYZ,Q,F)
C
C   Calculate x-component of vorticity FOR A 2D DATA as
C
C        du/dy-dv/dx
C
C
C
      DIMENSION XYZ(IDIM,JDIM,2),Q(IDIM,JDIM,2),F(IDIM,JDIM)
C
C   Loop through each point in the plane, calculating velocity derivatives and
C   metric terms using central differencing in the interior and dropping to
C   first order at the boundaries.
C
      DO 10 J= 1,JDIM
         DO 10 I= 1,IDIM
            IF (I.EQ.1) THEN
               UXI = Q(I+1,J,1)-Q(I,J,1)
               VXI = Q(I+1,J,2)-Q(I,J,2)
               XXI = XYZ(I+1,J,1)-XYZ(I,J,1)
               YXI = XYZ(I+1,J,2)-XYZ(I,J,2)
            ELSE IF (I.EQ.IDIM) THEN
               UXI = Q(I,J,1)-Q(I-1,J,1)
               VXI = Q(I,J,2)-Q(I-1,J,2)
               XXI = XYZ(I,J,1)-XYZ(I-1,J,1)
               YXI = XYZ(I,J,2)-XYZ(I-1,J,2)
            ELSE
               UXI = .5*(Q(I+1,J,1)-Q(I-1,J,1))
               VXI = .5*(Q(I+1,J,2)-Q(I-1,J,2))
               XXI = .5*(XYZ(I+1,J,1)-XYZ(I-1,J,1))
               YXI = .5*(XYZ(I+1,J,2)-XYZ(I-1,J,2))
            ENDIF
            IF (J.EQ.1) THEN
               UETA= Q(I,J+1,1)-Q(I,J,1)
               VETA= Q(I,J+1,2)-Q(I,J,2)
               XETA= XYZ(I,J+1,1)-XYZ(I,J,1)
               YETA= XYZ(I,J+1,2)-XYZ(I,J,2)
            ELSE IF (J.EQ.JDIM) THEN
               UETA= Q(I,J,1)-Q(I,J-1,1)
               VETA= Q(I,J,2)-Q(I,J-1,2)
               XETA= XYZ(I,J,1)-XYZ(I,J-1,1)
               YETA= XYZ(I,J,2)-XYZ(I,J-1,2)
```

```
          ELSE
             UETA= .5*(Q(I,J+1,1)-Q(I,J-1,1))
             VETA= .5*(Q(I,J+1,2)-Q(I,J-1,2))
             XETA= .5*(XYZ(I,J+1,1)-XYZ(I,J-1,1))
             YETA= .5*(XYZ(I,J+1,2)-XYZ(I,J-1,2))
          ENDIF
C
             F(I,J)= (-UXI*XETA+UETA*XXI)-(VXI*YETA-VETA*YXI)
   10     CONTINUE
C
      RETURN
      END
```